

Simulazione seconda prova Esame di Stato

Sito Web - Gestione di un centro agroalimentare all'ingrosso (Parte seconda)

Il candidato realizzi inoltre il sito Internet del centro che deve permettere, tra l'altro, di visualizzare tutti i dati dell'azienda del centro e la ricerca di una specifica azienda venditrice anche se l'utente del sito conosce soltanto una parte della sua denominazione.

Competenze verificate
<ul style="list-style-type: none"> ■ Progettare la struttura logica e fisica dei file di un sito Web ■ Codificare interrogazioni SQL su un database in pagine ASP ■ Costruire applicazioni Web interattive mediante pagine ASP

Volumi di riferimento
Teoria e laboratorio ICT Siti Web e soluzioni per l'e-commerce Modulo 1 Progetto di siti Web - UD 3 Ambienti software autore Modulo 2 Soluzioni per l'e-commerce – UD 2 Sviluppo di applicazioni Web

Riflessioni per il docente
<p>Il metodo di lavoro proposto ripercorre i passi per il progetto e la realizzazione di una soluzione per l'e-commerce, basata sulle pagine ASP il cui script è stato sviluppato con il linguaggio VBScript.</p> <p>Il sito Web è stato realizzato in FrontPage, per fornire la possibilità a tutti gli allievi di collaudare il sistema proposto. In allegato a questa prova è quindi possibile effettuare il download dell'intero sito Web, memorizzato nella cartella <i>centroingrosso</i>. Il sito Web è completo e funzionante per cui può essere collaudato pubblicandolo su un server Web di Internet oppure dell'intranet del laboratorio del proprio Istituto. In alternativa, il collaudo può essere realizzato su un qualsiasi computer stand-alone in cui è stato installato il programma di sistema che lo trasforma in un server Web. Esempi di programmi per creare un server Web sono: IIS (<i>Internet Information Services</i>) in un computer Windows NT/2000/XP, PWS (<i>Personal Web Server</i>) in un sistema Windows 95/98/Me e Apache in un sistema Linux. Le pagine ASP del sito accedono a un file di database di Access, per cui il server di distribuzione deve anche disporre dei driver per questo RDBMS.</p> <p>Per collaudare il sito Web possiamo seguire le tre fasi successive.</p> <p>1. Pubblicare la cartella <i>centroingrosso</i> effettuando l'upload su un server Web con una connessione HTTP (se il server Web dispone delle estensioni di FrontPage) oppure FTP. Nel caso di un server Web realizzato con un computer stand-alone, la pubblicazione può essere realizzata creando un nuovo sito Web vuoto di nome <i>http://NomeComputer/centroingrosso</i> oppure <i>http://localhost/centroingrosso</i> e importando (in FrontPage: comando <i>Importa...</i> del menu <i>File</i>) tutto il contenuto della cartella <i>centroingrosso</i> (file e singole cartelle) in quella del nuovo sito Web.</p> <p>Dopo aver svolto una delle operazioni precedenti, la home page del sito Web avrà come URL il seguente: http://NomeServerWeb/centroingrosso/default.htm</p> <p>2. Aggiungere alla cartella del sito Web con il database i permessi di scrittura necessari per permettere di effettuare operazioni di aggiornamento dei dati.</p> <p>Ad esempio, se si pubblica un sito su un server Web che dispone delle estensioni di FrontPage, è sufficiente accettare la creazione guidata della connessione al file di database proposta dal server. In questo caso, le estensioni di FrontPage memorizzano il file di database (<i>CentroIngresso.mdb</i>) in una cartella di nome <i>fpdb</i> che dispone automaticamente dei permessi sia di lettura sia di scrittura sul database stesso. In alternativa, per aggiungere il permesso di scrittura, si deve accedere (se si ha accesso al server di distribuzione) alla cartella del sito Web con il file di database e modificare manualmente i permessi mediante le funzioni locali del file system del sistema operativo.</p> <p>Il sito Web è stato realizzato in FrontPage 2003 ed è stato collaudato in un computer Windows (di nome <i>pc</i>) sia 2000 sia XP con installato il servizio IIS.</p>

Creazione del sito Web

Il sito è basato su una struttura a frame definita mediante il frameset *default.htm*, che rappresenta la home page. La **struttura logica** del sito è quella descritta nella figura seguente.




Il frameset è formato da tre frame con le seguenti funzioni:

- il frame superiore (*intestazione.htm*) è fisso e contiene il logo e la denominazione del centro all'ingrosso;
- il frame inferiore a sinistra (*menu.htm*) contiene il menu i cui hyperlink richiamano le due operazioni richieste del sistema informativo;
- la cornice inferiore a destra visualizza i risultati ottenuti dall'esecuzione delle operazioni.


Il codice sorgente HTML del frameset che descrive la struttura in frame della pagina *default.htm* è quindi il seguente.

```
<FRAMESET ROWS="14%,*" FRAMESPACING="0" BORDER="0" FRAMEBORDER="0" >
  <FRAME NAME="intestazione" SCROLLING="NO" NORESIZE TARGET="sommarrio"
    SRC="intestazione.htm">
  <FRAMESET COLS="27%,*" >
    <FRAME NAME="sommario" TARGET="principale" SRC="menu.htm"
      SCROLLING="AUTO" NORESIZE>
    <FRAME NAME="principale" SRC="start.htm" SCROLLING="AUTO">
  </FRAMESET>
</FRAMESET>
```

Il frame *menu.htm* contiene una barra di hyperlink che richiamano le pagine per l'avvio delle operazioni, sulla base della seguente struttura logica ad albero del sito Web.



La **struttura fisica** del sito Web è illustrata nella figura successiva.



Per realizzare l'applicazione distribuita client-server si è utilizzata la tecnologia software ASP (*Active Server Pages*), utilizzando come linguaggio di programmazione di script il VBScript.

Progetto e realizzazione della base di dati

Il database è quello realizzato nel RDBMS Access nella prima parte dell'esercitazione. Per rendere disponibile il database, dobbiamo quindi pubblicare il file *DBCentroIngresso.mdb*, ad esempio, nella sottocartella *fpdb* del sito Web *centroingrosso*.

Impostazione della connessione del sito Web al database

Per l'accesso a una base di dati, il VBScript utilizza la tecnologia orientata agli oggetti ADO e le interfacce software basate sugli standard OLE DB e ODBC. Per collegare una pagina ASP con il database dobbiamo definire una nuova connessione logica, ad esempio di nome *ConnessioneDB*, usando le seguenti istruzioni VBScript.

```
Dim ConnessioneDB
Set ConnessioneDB = Server.CreateObject("ADODB.Connection")
```

Nel modello ADO, la connessione *ConnessioneDB* è un oggetto della classe *Connection*. Se seguiamo lo standard OLE DB, l'oggetto *ConnessioneDB* deve essere collegato al file fisico di database mediante il metodo *Open* (della classe *Connection*) e una stringa di connessione, con una sintassi del tipo che segue.

```
ConnessioneDB.Open StringaConnessione
```

Nel nostro progetto, con pagine ASP in sottocartelle della radice (*root*) del sito Web, la stringa di connessione per il RDBMS Access assume quindi una forma del tipo presentato nel seguito.

```
StringaConnessione = "DRIVER={Microsoft Access Driver (*.mdb)}; " & _
"DBQ=" & Server.MapPath("../fpdb/DBCentroIngresso.mdb")
```

Il metodo **MapPath** dell'oggetto predefinito *Server* usato con la sintassi seguente, restituisce il percorso fisico del server Web dove è effettivamente memorizzato il file con il database.

```
Server.MapPath("URL Relativo File")
```

Utilizzando *Server.MapPath* nella stringa di connessione è quindi possibile rendere indipendente la connessione logica a un file di database dalla sua effettiva posizione sulle memorie di massa del server Web di distribuzione.

Nell'URL relativo per il file di database si sono usati i simboli ".." per individuare la radice del sito Web da una sua sottocartella. Le nuove versioni del software di sistema (quale IIS 6), che realizzano il servizio WWW, non supportano i simboli ".." che sono sostituiti da uno slash "/" per indicare la radice del sito. Nel caso di questi server Web, la stringa di connessione diventa la seguente.

```
StringaConnessione = "DRIVER={Microsoft Access Driver (*.mdb)}; " & _  
"DBQ=" & Server.MapPath( "/fpdb/DBCentroIngresso.mdb" )
```

Nell'applicazione Web, in fase di realizzazione, useremo una delle stringhe di connessione precedenti (sulla base della versione del software di sistema del server), in modo che il sito possa essere pubblicato su qualsiasi server Web di distribuzione senza effettuare alcuna modifica al codice sorgente dello script.

Dichiarazione della stringa di connessione al database come variabile globale

La stringa di connessione può essere dichiarata come variabile globale in modo che sia visibile durante una sessione HTTP di un utente all'applicazione Web. Per questo obiettivo è sufficiente creare un file di testo di nome *global.asa*, nella radice del sito Web, e dichiarare la variabile *StringaConnessione* nel corpo dell'evento *Session_OnStart()* con la seguente sintassi.

global.asa

```
<SCRIPT LANGUAGE="VBScript" RUNAT="SERVER">  
Sub Session_OnStart()  
    Dim StringaConnessione  
    Session("StringaConnessione")="DRIVER={Microsoft Access Driver (*.mdb)}; " & _  
        "DBQ=" & Server.MapPath( "../fpdb/DBCentroIngresso.mdb" )  
End Sub  
</SCRIPT>
```

Il file *global.asa* può essere già presente nel sito Web; in questo caso, è sufficiente modificare il contenuto aggiungendo, nella procedura di evento *Session_OnStart()*, la dichiarazione della variabile globale *StringaConnessione*.

Una variabile globale è visibile in tutte le pagine del sito Web relative a una sessione, utilizzando l'identificatore: *Session("VariabileGlobale")*.

Nel nostro esempio, dopo avere dichiarato la variabile globale *StringaConnessione*, per creare la connessione con il file di database possiamo impiegare, in tutte le pagine *.asp* di accesso ai dati, la seguente istruzione:

```
ConnessioneDB.Open Session("StringaConnessione")
```

L'uso di una variabile globale può essere utile nel caso di applicazioni Web con molte pagine di accesso ai dati, infatti, nel caso di modifica del percorso del file di database sul server Web, è sufficiente riscrivere la stringa di connessione una sola volta, nel file *global.asa*, e non in tutte le pagine *.asp* che impiegano la connessione stessa.

Sviluppo dell'applicazione client-server distribuita

Operazione: aziende del centro

Per realizzare questa operazione, possiamo progettare la pagina ASP *elenco_aziendecentro.asp*, richiamata nel menu del sito Web con l'hyperlink [Elenco aziende del centro](#). Lo script (riportato nel seguito) inserito nella pagina ASP esegue la query per la ricerca di tutte le aziende del centro sul database e presenta una tabella con i risultati visualizzati in una pagina HTML dinamica di risposta.

```
servizi/elenco_aziendecentro.asp
```

```
<%@ Language=VBScript %>
<% Response.Buffer=True %>

<HTML>
<HEAD><TITLE>Ricerca dei dati di tutte le aziende del centro </TITLE></HEAD>
<BODY>
<H3>Elenco di tutte le aziende del centro all'ingrosso</H3>
<% Dim ConnessioneDB, rsTabella, Query
   Set ConnessioneDB = Server.CreateObject("ADODB.Connection")
   ConnessioneDB.Open Session("StringaConnessione")
   Query = "SELECT * " &_
           " FROM AziendeCentro " &_
           " ORDER BY Denominazione ASC;"
   Set rsTabella = ConnessioneDB.Execute(Query)%>
<% If rsTabella.BOF=True And rsTabella.EOF=True Then %>
  <H4>Non ci sono aziende disponibili </H4>
<% Else %>
  <TABLE BORDER="1">
  <TR>
    <TH>Spazio vendita</TH><TH>Denominazione</TH><TH>Dati titolare</TH>
    <TH>Telefono</TH><TH>Telefono centro</TH>
  </TR>
  <% Do until rsTabella.EOF=True %>
    <TR>
      <TD ALIGN="CENTER"><%=rsTabella.Fields("SpazioVendita")%></TD>
      <TD ALIGN="CENTER"><%=rsTabella.Fields("Denominazione")%></TD>
      <TD ALIGN="CENTER"><%=rsTabella.Fields("Titolare")%></TD>
      <TD ALIGN="CENTER"><%=rsTabella.Fields("Telefono")%></TD>
      <TD ALIGN="CENTER"><%=rsTabella.Fields("TelefonoCentro")%></TD>
    </TR>
    <% rsTabella.MoveNext %>
  <% Loop %>
  </TABLE>
<% End If %>

<% rsTabella.Close : Set rsTabella = Nothing
   ConnessioneDB.Close : Set ConnessioneDB = Nothing
   Session.Abandon %>
</BODY>
</HTML>
```

Operazione: ricerca azienda del centro

Per realizzare questa operazione, possiamo progettare la pagina HTML *modulo_aziendacentro.htm*, richiamata nel menu del sito Web con l'hyperlink [Ricerca azienda del centro](#), che contiene (nel suo BODY) un modulo (vedi figura e codice HTML successivi) per la raccolta del parametro per l'esecuzione del servizio.

Servizio: ricerca dei dati di un'azienda del centro all'ingrosso

Denominazione dell'azienda (o una sua parte)

↑
txtDenominazione

```
<FORM ACTION="ricerca_aziendacentro.asp" METHOD="POST">
  <FIELDSET>
    <LEGEND><B>Servizio: ricerca dei dati di un'azienda del centro
      all'ingrosso</B></LEGEND>
    Denominazione dell'azienda (o una sua parte)
    <INPUT TYPE="TEXT" NAME="txtDenominazione" SIZE="30"> <BR><BR>
    <INPUT TYPE="SUBMIT" VALUE="Invia" NAME="cmdInvia">
    <INPUT TYPE="RESET" VALUE="Reimposta" NAME="cmdReimposta">
  </FIELDSET>
</FORM>
```

Il comando del form per l'input dei parametri del servizio

```
<FORM ACTION="ricerca_aziendacentro.asp" METHOD="POST">
```

richiama la pagina *ricerca_aziendacentro.asp*, il cui script (riportato nel seguito) esegue la query sul database e presenta una tabella con i risultati visualizzati in una pagina HTML dinamica di risposta. Usando il valore "POST" per l'attributo METHOD del tag doppio FORM, i dati del modulo sono inviati alla pagina *.asp* all'interno del corpo di un messaggio HTTP.

```
servizi/ricerca_aziendacentro.asp
<%@ Language=VBScript %>
<% Response.Buffer=True %>

<HTML>
<HEAD><TITLE>Ricerca dei dati di un'azienda del centro </TITLE></HEAD>
<BODY>
<H3>Dati dell'azienda del centro ricercati</H3>
<% Dim ConnessioneDB, rsTabella, Query, Denominazione
  Set ConnessioneDB = Server.CreateObject("ADODB.Connection")
  ConnessioneDB.Open Session("StringaConnessione")

  Denominazione = Request.Form("txtDenominazione")

  Query="SELECT * " &_
        " FROM AziendeCentro " &_
        " WHERE Denominazione LIKE '%" & Denominazione & "%'; "

  Set rsTabella = ConnessioneDB.Execute(Query)%>
<% If rsTabella.BOF=True And rsTabella.EOF=True Then %>
  <H4>L'azienda cercata non esiste</H4>
<% Else %>
```

```
<TABLE BORDER="1">
<TR>
  <TH>Spazio vendita</TH><TH>Denominazione</TH><TH>Dati titolare</TH>
  <TH>Telefono</TH><TH>Telefono centro</TH>
</TR>
<% Do until rsTabella.EOF=True %>
  <TR>
    <TD ALIGN="CENTER"><%=rsTabella.Fields("SpazioVendita")%></TD>
    <TD ALIGN="CENTER"><%=rsTabella.Fields("Denominazione")%></TD>
    <TD ALIGN="CENTER"><%=rsTabella.Fields("Titolare")%></TD>
    <TD ALIGN="CENTER"><%=rsTabella.Fields("Telefono")%></TD>
    <TD ALIGN="CENTER"><%=rsTabella.Fields("TelefonoCentro")%></TD>
  </TR>
  <% rsTabella.MoveNext %>
<% Loop %>
</TABLE>
<% End If %>

<% rsTabella.Close : Set rsTabella = Nothing
ConnessioneDB.Close : Set ConnessioneDB = Nothing
Session.Abandon %>
</BODY>
</HTML>
```

Osserviamo che nel codice della query eseguita tramite il servizio HTTP:

```
Query="SELECT * " & _
      " FROM AziendeCentro " & _
      " WHERE Denominazione LIKE '%" & Denominazione & "%'; "
```

il codice jolly * è stato sostituito con quello % standard dell'SQL.